
Foreword 

 

The purpose of this guidebook is to help you successfully use SAS 9.4 to complete the course 

assignments for EPRS 8530: Quantitative Methods & Analysis I, EPRS 8540: Quantitative 

Methods & Analysis II, and EPRS 8550: Quantitative Methods & Analysis III.  Specifically, 

this guide will provide assistance as you (1) analyze data using a variety of statistical models, 

(2) verify the tenability of the associated statistical assumptions, (3) read and interpret the 

corresponding output, and (4) report your results in an appropriate format and style. 

 

For each method of statistical analysis, you will find the following: 

1. a sample research scenario with a small dataset, 

2. SAS code for executing the analysis with explanatory notes and tips, and 

3. excerpts of selected output noting the results most commonly of interest to the 

researcher, and 

4. sample summary statements of the results following American Psychological 

Association (APA) 6th Edition reporting guidelines. 

 

Please note that most of the scenarios and datasets herein are intentionally small and simplistic, 

so that you can focus your attention on the SAS code and output presented.  The code needed to 

run analyses will invariably differ for your own particular course assignments and research 

projects.  Pay careful attention to the explanatory notes provided, as these should help you 

adapt the code to your specific needs. 

 

It is our hope that you will find this guidebook helpful as you learn new statistical methods and 

that it also comes in handy as a reference for you later in your statistics career. 

 

 

Theresa Dell-Ross 

Department of Educational Policy Studies 

Georgia State University 

 

T. Chris Oshima 

Department of Educational Policy Studies 

Georgia State University 

About This Guidebook 

 

This guidebook was developed specifically for students taking statistical courses in the 

Educational Policy Studies Department of the College of Education and Human Development at 

Georgia State University. Anyone conducting research analysis who is new to using SAS may 

find this guide helpful. 

 

This guidebook is organized in three main sections. The first section, Getting Started, provides a 

brief introduction to SAS 9.4.  This section includes directions for you to install SAS; write, 

run, and save code; check your code for execution errors; and “clean” your SAS windows.  The 

second section, Descriptive Statistics, will provide instruction related to the generation of 

measures of central tendency, measures of spread, Pearson’s r for correlation, and frequency 

tables and histograms.  In the third section, Inferential Statistics, you will learn to run analyses 

ranging from a one-sample t test to multiple regression and model building with the 

incorporation of assumption-checking methods. 

 

The code herein was written using SAS 9.4, with default installation settings, in a Windows 10 

environment; it is assumed that the reader is using this version of SAS (or a more current one) 

in Windows 10 or higher. 

 

It is assumed that students already possess a basic knowledge of Windows and standard 

computer software (e.g., word processing programs). A working knowledge of any 

programming language will be helpful, but is not required.  As the guidebook progresses from 

one analysis to another, certain levels of detail are omitted due to space limitations. (For 

example, explanation of the assignment of raw data to variable names need not be repeated for 

each successive statistical analysis once it has been introduced.)  It is expected that the user 

will develop a cumulative working knowledge of SAS as the guidebook progresses. 

 

The SAS code presented herein is limited to the procedures required for the statistical analyses 

being conducted.  (For example, topics such as the manipulation of character data are omitted.)  

Of course, there are often multiple methods for achieving the same results; however, in most 

cases, only one way is presented.  Readers wishing to learn SAS in depth are therefore referred 

to SAS courses provided by the SAS Institute, online resources, or print resources. 

 

Every effort has been made to ensure that this guidebook is free from errors. If you should find 

any mistakes within this guide, please contact Theresa Dell-Ross (tdellross1@student.gsu.edu) 

or Dr. Chris Oshima (oshima@gsu.edu) so that corrections may be made. 

 

Please note the following: SAS 9.4 is copyrighted 2012 by the SAS Institute, Inc. and Microsoft 

Windows® is a registered trademark of the Microsoft Corporation. Users of this guidebook are 

expected to obey all copyright laws of the United States, including software licensing 

agreements.  

mailto:tdellross1@student.gsu.edu
mailto:oshima@gsu.edu

Getting Started with SAS 9.4 

 

Installing SAS 

 

SAS 9.4 is currently offered for free to Georgia State University students.  To download and 

install SAS, go to the GSU Technology home page at https://technology.gsu.edu/technology-

services/services-for-you/it-services-for-students/.  Scroll down to the “Download Software” 

section, find SAS, and follow the directions. 

 

Using SAS 

 

When you open SAS, this is what you will see. 

 
 

The “Editor” window is where you will write and run your code.  The “Log” window is where 

you will check to make sure that your code was executed properly.  You can toggle back and 

forth using the Editor and Log tabs at the bottom of the SAS screen. 

 

You may begin typing code in the Editor window.  You may also open an existing code file by 

going to File > Open.  To save your code, make sure you are in the Editor window and then go 

to File > Save As or File > Save.  To run your code, highlight it and click on the icon of the 

running man. 

 

 
 

Here is what you will see once you have written and run some code. 

 

https://technology.gsu.edu/technology-services/services-for-you/it-services-for-students/
https://technology.gsu.edu/technology-services/services-for-you/it-services-for-students/

Notice that there is now a list of results in the left-hand window.  There is also a “Results” 

window (with a Results tab at the bottom). 

 

IT IS HIGHLY RECOMMENDED THAT YOU GET IN THE HABIT OF READING THE 

LOG WINDOW IMMEDIATELY AFTER YOU RUN CODE, EVERY TIME YOU RUN 

CODE, BEFORE YOU LOOK AT ANY OF THE RESULTS.  It is possible that SAS will 

generate results, even if you made mistakes in your code, and you do not want to interpret or 

report results without first checking to make sure that the code was executed properly. 

 

IN SAS, COLOR IS CRITICAL.  When you type in your code, SAS automatically color-codes 

certain words.  As you can see from the screenshot below, it colors some words in dark blue, 

light blue, green, teal, purple, and black.  As you use this guide, make sure your colors match 

the colors you see in the code here.  If your colors don’t match the colors here, it is a clue that 

you may have made an error (e.g., forgetting a semicolon in the line above or misspelling a 

word). 

COLOR IS ALSO CRITICAL IN THE LOG WINDOW.  When you check the Log window, 

you always want to see black and blue.  The Log window will show your code in black.  If there 

are any errors, they will be shown in maroon (red).  If there are warnings, they will be shown in 

green.  This screenshot shows an error and two warnings: 

 

 
 

All other notes will be shown in blue.  In most cases, blue means success.  However, you must 

be careful.  Blue does not mean success 100% of the time.  For example, look at the following:  

Notice that “178 records were read” but the new dataset has “172 observations.”  There was, in 

fact, an error here (having to do with missing data in the .CSV file that was being read in), but 

SAS did not recognize it as such.  ALWAYS read the Log window before interpreting and 

reporting results!! 

 

Keeping Your Windows Manageable 

 

Sometimes you may run the same code over and over as you troubleshoot, and you may become 

overloaded with output in Results window and notes in the Log window.  It is easy to fix this.  

If you want to clear everything out, simply do the following. 

1. In the left-hand window, click on the word “Results” at the top of the list of all the 

output.  Click “Delete” on your keyboard and answer “Yes.” 

2. Select the “Log” window tab, go to “Edit,” and select “Clear all.” 

  
3. Continue working.  

Descriptive Statistics 

Research Scenario 

 

A counselor working with a group of caregivers of patients living with a terminal illness is 

interested in forming a support group to share experiences and therefore reduce the sense of 

isolation often associated with catastrophic illness. The counselor, working with hospital staff, 

administers a depression and anxiety inventory to each caregiver that has volunteered for the 

program. The counselor feels that knowing the levels of depression and anxiety within the 

group will help in the design of an effective intervention program. The scores obtained from the 

administration of the two inventories are given below. 

 

ANXIETY DEPRESSION 

22 16 

12 8 

68 33 

10 6 

5 5 

53 24 

44 18 

37 17 

0 2 

21 14 

64 31 

33 17 

55 30 

18 13 

3 3 

4 4 

11 7 

13 9 

7 5 

Descriptive Statistics  

SAS Code for Measures of Central Tendency, Measures of Spread, and 

Correlation 

 

(1) DATA CAREGIVER; 

(2)  INPUT ANXIETY DEPRESSION; 

(3)  LINES; 

   22 16 

   12 8 

   68 33 

   10 6 

   5 5 

   53 24 

   44 18 

   37 17 

   0 2 

   21 14 

   64 31 

   33 17 

   55 30 

   18 13 

   3 3 

   4 4 

   11 7 

   13 9 

   7 5 

(4) RUN;  

 

(5) PROC PRINT DATA=CAREGIVER; 

(6) PROC MEANS DATA=CAREGIVER; 

(7) PROC CORR DATA=CAREGIVER PLOTS=SCATTER(ELLIPSE=NONE); 

(8) RUN; 

(9) QUIT; 

 

(1) The DATA step creates a new dataset and assigns it the specified name, CAREGIVER in 

this case. 

(2) The INPUT statement creates the variable names and assigns the order of the variables to 

the new dataset.  This command also assigns a variable type to each new variable.  The 

default variable type, which applies to both variables in this case, is numeric.  If a 

character variable is being created, put a dollar sign ($) in back of it.  For example, if 

ANXIETY was a character variable, the code would read INPUT ANXIETY $ 

DEPRESSION. 

(3) LINES indicates that the data are being entered manually (as opposed to read in or 

imported from a file).  The data are listed beginning on the next line.  Make sure that you 

enter the data in the same order that you entered the variable names in the INPUT 

statement (i.e. type the anxiety score first, followed by the depression score, because 

ANXIETY comes first and DEPRESSION comes second).  Please note that the data do 

NOT include a semicolon (;) at the end. 

(4) This RUN statement is optional, because SAS automatically ends the step that is currently 

running each time SAS encounters a new DATA or PROC step.  Please note that, for 

some mysterious reason, this RUN statement does not turn dark blue like other RUN 

statements.  😊 

(5) PROC PRINT is the “proc”edure that returns all of the raw data from CAREGIVER to 

your output window.  Making use of this procedure is optional, but it’s recommended 

when you enter data manually; it is a built-in check for data entry errors. 

(6) PROC MEANS is the procedure that outputs the sample size, mean, standard deviation, 

minimum, and maximum for each numeric variable in CAREGIVER. 

(7) PROC CORR is the procedure that runs the correlations for each pair of numeric variables 

in CAREGIVER, as well as the p value for each correlation.  It also returns the same 

descriptive statistics as PROC MEANS (rendering that procedure redundant).  In this 

procedure, a scatterplot is also requested, with the default prediction ellipses omitted. 

(8) Each time SAS encounters a new DATA or PROC step, SAS automatically ends the step 

that is currently running.  Because there are no more DATA or PROC steps in this code, 

the RUN step is needed in order to force SAS to end the PROC CORR step, effectively 

ending this code.  The RUN step can be used at the end of any DATA or PROC step, not 

just the last one in the code.  It is generally good practice to include RUN at the end of 

each DATA or PROC step. 

(9) In SAS, some procedures will not end using RUN alone; a QUIT step is also required.  If 

you forget to include the QUIT step, and SAS shows a procedure as still running at the top 

of the window, you can simply type “QUIT;” (without the quotation marks), then 

highlight and run this code; it will end any procedure still running.  It is simpler – and, 

therefore, recommended – to get in the habit of including a QUIT statement at the end of 

your code.  

Descriptive Statistics  

SAS Code for Frequency 

 
(10) PROC FORMAT; 

(11)  VALUE ANX_FMT 

(12)   0-9 = '0-9' 

   10-19 = '10-19' 

   20-29 = '20-29' 

   30-39 = '30-39' 

   40-49 = '40-49' 

   50-59 = '50-59' 

   60-69 = '60-69'; 

(13)  VALUE DEP_FMT 

   0-4 = '0-4' 

   5-9 = '5-9' 

   10-14 = '10-14' 

   15-19 = '15-19' 

   20-24 = '20-24' 

   25-29 = '25-29' 

   30-34 = '30-34'; 

RUN; 

 

(14) PROC FREQ DATA=CAREGIVER; 

(15)  FORMAT ANXIETY ANX_FMT.; 

(16)  FORMAT DEPRESSION DEP_FMT.; 

RUN; 

 

(17) PROC UNIVARIATE DATA=CAREGIVER; 

(18)  VAR ANXIETY; 

(19)  HISTOGRAM ANXIETY /  

(20)   VSCALE=COUNT 

(21)   ENDPOINTS = 0 TO 70 BY 10 

(22)   ODSTITLE='Frequency of Anxiety Scores' 

(23)   VAXISLABEL='Frequency'; 

(24)  LABEL ANXIETY='Anxiety'; 

RUN; 

 

(25) PROC UNIVARIATE DATA=CAREGIVER; 

  VAR DEPRESSION; 

  HISTOGRAM DEPRESSION /  

   VSCALE=COUNT 

   ENDPOINTS = 0 TO 35 BY 5 

   ODSTITLE='Frequency of Depression Scores' 

   VAXISLABEL='Frequency'; 

  LABEL DEPRESSION='Depression'; 

(26) RUN; 

(27) QUIT; 

 

(10) PROC FORMAT is used to create guidelines for formatting a variable.  In this case, the 

procedure will create a format that specifies the ranges, or bins, for the frequency analysis.  

Your bins should always include the minimum observed value for the variable, the 

maximum observed value, and all values in between even if unobserved.  For example, if 

your variable X has values of 1, 2, 3, 4, and 10, you need to make sure that (a) your lowest 

bin includes the value 1, (b) your highest bin includes the value 10, and (c) you have bins 

for 5, 6, 7, 8, and 9, even though X does not include these values.  If you omit bins for 5-9, 

your X data will be misrepresented. 

(11) The VALUE statement assigns a name to the format you are creating.  In this case, we are 

creating the formatting for ANXIETY, which is reflected in the name assigned here 

(ANX_FMT). 

(12) Beginning on this line, each bin is created and assigned a label.  Values of 0-9 (on the left 

side of the equal sign), for example, are assigned a label of “0-9” (on the right side of the 

equal sign).  Because the label is contained inside quotation marks, you can include any 

character in the label (e.g., <, >, $, letters/words, etc.).  For example, you could write: 0-9 

= ‘≤ 9’.  ANX_FMT has no impact on your output yet, because it has not been applied to a 

particular variable. 

(13) The VALUE statement assigns a name to the format you are creating.  In this case, we are 

creating the formatting for DEPRESSION, which is reflected in the name assigned here 

(DEP_FMT).  Following this statement, you will see the creation and labeling of the bins, 

as in (12). 

(14) PROC FREQ generates a frequency table for each numeric variable in CAREGIVER. 

(15) The FORMAT statement tells SAS that you want a frequency table, but that you need to 

override the default frequency table format.  The default for numeric variables is to 

include a bin for every single observed value.  As we want the frequency by ranges, this 

statement is used to apply the ANX_FMT range values and labels to ANXIETY for this 

particular analysis.  (It does NOT change the actual ANXIETY data.)  Note that a period 

(.) follows ANX_FMT.  The code will not work properly if you omit the period. 

(16) Just as in (15), the FORMAT statement overrides the default frequency table, applying the 

DEP_FMT values and labels to DEPRESSION.  Note that a period (.) follows DEP_FMT.  

The code will not work properly if you omit the period. 

(17) PROC UNIVARIATE provides descriptive statistics for each variable, and renders PROC 

MEANS redundant.  In fact, PROC UNIVARIATE provides a greater variety of 

descriptive statistics than PROC MEANS, including median, mode, skewness, and 

kurtosis.  PROC UNIVARIATE is used here to generate histograms. 

(18) The VAR statement limits the analysis in the PROC UNIVARIATE step to ANXIETY 

(i.e. DEPRESSION is omitted). 

(19) The HISTOGRAM statement creates histograms for the specified variable(s).  Here, a 

histogram is created for ANXIETY.  The forward slash (/) is used to separate the call for a 

histogram of ANXIETY from the options that follow. 

(20) VSCALE is the first option in relation to the histogram being created.  The default scale 

for the vertical axis is the frequency percent (e.g., a particular value occurs 20% of the 

time).  VSCALE overrides this and sets the vertical axis to the frequency count (e.g., a 

particular value occurs 5 times). 

(21) The ENDPOINTS option sets the bin ranges, in this case from 0 to 70 in increments of 10.  

(This matches ANX_FMT.) 

(22) The ODSTITLE option assigns a title to the histogram. 

(23) The VAXISLABEL option overrides the default vertical axis label (“Count”) and assigns 

a different label.  This is optional; you can always use the default label.  As this is the last 

option in the code, it ends with a semicolon (;). 

(24) In order to change the horizontal axis label, the LABEL statement is used.  This is 

optional; the default is the name of the variable being represented in the histogram 

(ANXIETY in this case).  

(25) Here, PROC UNIVARIATE is used to create a histogram for DEPRESSION.  The code 

matches that of the PROC UNIVARIATE code for ANXIETY, except for the variable 

included (DEPRESSION in this case), and title, labels, and bin ranges of the histogram.  

(The ENDPOINTS option sets the bin ranges from 0 to 35 in increments of 5; this matches 

DEP_FMT.) 

(26) Remember that it is generally good practice to include RUN at the end of each DATA or 

PROC step; in this code it is used to end PROC FORMAT, PROC FREQ, and PROC 

UNIVARIATE.  You may decide not to use RUN in the first two instances – although this 

is NOT recommended – but you MUST use it at the end of the code. 

(27) Remember that some procedures will not end using RUN alone; a QUIT step is also 

required.  Therefore, you should include it at the end of every code. 

Descriptive Statistics  

Selected Output 

Inferential Statistics 

One-Sample t Test 

Research Scenario 

 

Suppose that Professor Coffey learns from a national survey that the average high school 

student in the United States spends 6.75 hours each week exploring particular websites on the 

internet.  The professor is interested in knowing how internet use among students at the local 

high school compares with this national average.  Is local use more than, or less than, this 

average?  Professor Coffey randomly selects a sample of 10 students.  Each student is asked to 

report the number of hours he or she spends exploring these websites on the internet in a typical 

week during the school year.  The data appear below. 

 

Student 
Number of 

Hours 

1 6 

2 9 

3 12 

4 3 

5 11 

6 10 

7 18 

8 9 

9 13 

10 8 

 

Source of Data and Scenario: Coladarci, T., & Cobb, C. (2014). Fundamentals of Statistical 

Reasoning in Education (4th ed.). Hoboken, NJ: John Wiley & Sons, Inc.  

Inferential Statistics 

One-Sample t Test  

SAS Code 

 
DATA STUDYHOURS; 

  INPUT HOURS; 

  LINES; 

   6 

   9 

   12 

   3 

   11 

   10 

   18 

   9 

   13 

   8 

RUN;  

 

PROC PRINT DATA=STUDYHOURS; 

RUN; 

 

PROC UNIVARIATE DATA=STUDYHOURS; 

  VAR HOURS; 

  HISTOGRAM HOURS /  

   VSCALE=COUNT 

   ENDPOINTS = 0 TO 20 BY 1 

   ODSTITLE='Frequency of Study Hours' 

   VAXISLABEL='Frequency'; 

  LABEL HOURS='Hours'; 

RUN; 

 

(1) PROC TTEST DATA=STUDYHOURS PLOTS(shownull)=interval H0=6.75; 

(2)  VAR HOURS; 

(3)  TITLE "One-Sample t Test: Study Hours"; 

RUN; 

 

(4) TITLE; 

QUIT; 

 

(1) PROC TTEST may be used for one-sample t tests, as well as independent and dependent t 

tests. The PLOTS(shownull) option will add a reference line for the null hypothesis value 

(i.e. the national average of 6.75).  Setting “H0” to 6.75 will make this a one-sample t test. 

(2) The one-sample t test is being conducted for the variable HOURS. 

(3) An optional title is being assigned to the output.  Although this is not necessary here, it 

becomes very helpful in more advanced analyses that contain a lot of output, which may 

easily become confusing.  If you do not want a title, simply remove this line of code. 

(4) The title that was created in (3) will be applied to all future SAS output indefinitely.  For 

example, if you were to run PROC MEANS or PROC FREQ after this code, it would label 

the output “One-Sample t Test: Study Hours,” even though they have nothing to do with t 

tests.  You must take one of three explicit actions to clear this title: (a) exit and restart 

SAS, (b) assign a new title, or (c) type “TITLE;” (without the quotation marks) to reset the 

title to SAS defaults.  Option C is being enacted here to reset default titles. 

Inferential Statistics 

One-Sample t Test  

Selected Output 

Inferential Statistics 

Independent t Test 

Research Scenario 

 

A training manager believes that a new interactive computer-based training package will help 

improve the production rate of order assemblers. She arranges for a production area of 21 

experienced employees to complete the new training package over a six-week period. Another 

group from the production area of 23 employees received no additional training. The following 

are the average production rates per person per hour based on a 12-week period following the 

training. 

 

Training Group (T) Control Group (C) 

26 44 

45 45 

60 57 

73 28 

45 64 

51 39 

63 35 

46 43 

69 21 

51 56 

55 22 

58 87 

61 48 

54 12 

64 19 

56 62 

59 55 

35 44 

48 39 

45 44 

59 57 

-- 30 

-- 30 

Inferential Statistics 

Independent t Test  

SAS Code 

 
(1) PROC FORMAT; 

(2)  VALUE $TRTMNT_FMT 

(3)   "T"="Training" 

   "C"="Control"; 

RUN; 

 

DATA PRODUCTION; 

(4)  INPUT TREATMENT $ PRODRATE @@; 

(5)  FORMAT TREATMENT $TRTMNT_FMT.; 

  LINES; 

   T 26 T 45 T 60 T 73 T 45 

   T 51 T 63 T 46 T 69 T 51 

   T 55 T 58 T 61 T 54 T 64 

   T 56 T 59 T 35 T 48 T 45 

   T 59 C 44 C 45 C 57 C 28 

   C 64 C 39 C 35 C 43 C 21 

   C 56 C 22 C 87 C 48 C 12 

   C 19 C 62 C 55 C 44 C 39 

   C 44 C 57 C 30 C 30 

RUN; 

 

PROC PRINT DATA=PRODUCTION; 

RUN; 

 

(6) PROC GLM DATA=PRODUCTION PLOTS=DIAGNOSTICS; 

(7)  CLASS TREATMENT; 

(8)  MODEL PRODRATE=TREATMENT; 

(9)  MEANS TREATMENT / HOVTEST=LEVENE(TYPE=ABS); 

(10)  TITLE "Independent t Test: Production Rate"; 

RUN; 

 

(11) TITLE; 

QUIT; 

 

(1) The PROC FORMAT step is used here to create labels for the TREATMENT codes. 

(2) The VALUE statement assigns a name to the format you are creating, TRTMNT_FMT in 

this case, and matches each code with a label.  The dollar sign ($) at the front of 

TRTMNT_FMT indicates that this format will be applied to a character variable.  (If the $ 

is absent, SAS defaults to applying the format to a numeric variable.)  Remember to use 

quotation marks (“”) with character data and labels. 

(3) Beginning on this line, individual labels are assigned.  For example, “T” is assigned the 

label “Training.”  TRTMNT_FMT has no impact on your output yet, because it has not 

been applied to a particular variable. 

(4) The INPUT statement creates a character variable TREATMENT (the $ that follows it 

indicates that it is a character variable) and a numeric variable PRODRATE (the lack of $ 

means that it defaults to numeric).  TREATMENT is coded as “T” for employees who 

participated in the training package and “C” for those in the control group.  PRODRATE 

is the rate of production.  The @@ symbols tell SAS that there are multiple observations 

on a single line of data.  For example, the first line of data (T 26 T 45 T 60 T 73 T 45) 

contains the data for five observations (i.e. employees).  This is optional, but tends to save 

space in longer datasets. 

(5) The FORMAT statement is used to apply the labels created in TRTMNT_FMT (1) to 

TREATMENT.  Note that a period (.) follows TRTMNT_FMT; the code will not work 

properly if you omit the period. 

(6) PROC GLM (General Linear Model) is used to conduct regression analysis; a t test is one 

specific type of regression analysis.  In this case, an independent t test is being conducted 

on the PRODUCTION data.  The diagnostic plots have also been requested here; these 

will be used to verify the tenability of the t test assumptions (i.e. that the assumptions have 

been met).  Specifically, these plots will help when verifying that the data are normally 

distributed. 

(7) The CLASS statement is used to identify variables with nominal data (as opposed to 

continuous data); in other words, use a CLASS statement to identify categorical variables.  

In this case, you have the categorical/grouping variable TREATMENT. 

(8) The MODEL statement is written as Dependent Variable (DV) = Independent Variable 

(IV).  Again, the DV goes on the left of the equal sign and the IV goes on the right. 

(9) The MEANS statement requires SAS to compute the means for each group in 

TREATMENT.  The option HOVTEST=LEVENE calls for the Levene’s Test of 

Homogeneity of Variances, to test the assumption that the groups have equal variances.  

There are multiple homogeneity tests available; Levene’s Test is the default.  There are 

two methods for calculating Levene’s statistic: using the absolute residuals (TYPE=ABS) 

or the squared residuals (TYPE=SQUARE, the default).  As the users of this guide may be 

more familiar with SPSS or may be referencing SPSS: A Survival Guide for EPRS 8530 

and EPRS 8540, the TYPE option is set to the method used by SPSS: ABS. 

(10) An optional title is being assigned to the output.  Although this is not necessary here, it 

becomes very helpful in more advanced analyses that contain a lot of output, which may 

easily become confusing.  If you do not want a title, simply remove this line of code. 

(11) The title that was created in (10) will be applied to all future SAS output indefinitely.  For 

example, if you were to run PROC MEANS or PROC FREQ after this code, it would label 

the output “Independent t Test: Production Rate,” even though they have nothing to do 

with t tests.  You must take one of three explicit actions to clear this title: (a) exit and 

restart SAS, (b) assign a new title, or (c) type “TITLE;” (without the quotation marks) to 

reset the title to SAS defaults.  Option C is being enacted here to reset default titles. 

Inferential Statistics 

Independent t Test  

Selected Output 

 

 
 



R 3.4.1: A Survival Guide 

 

37 

 

 
 

 
 



R 3.4.1: A Survival Guide 

 

38 

 

 
 

 
 



R 3.4.1: A Survival Guide 

 

39 

 

 
  



R 3.4.1: A Survival Guide 

 

40 

 

Inferential Statistics 

Dependent t Test  

Research Scenario 

 

A sports psychologist was interested in testing the effect of a simple relaxation technique on 

college basketball players’ free throw shooting accuracy. Each player was asked to shoot 20 

consecutive free throws and the number of successful attempts was recorded. The players were 

then trained to use a simple five-second relaxation technique while preparing to shoot a free 

throw. The players then returned to the court and shot 20 consecutive free throws again. The 

resulting data are given below. 

 

Pre-Training Post-Training 

12 13 

15 15 

9 11 

16 15 

12 15 

15 18 

17 17 

10 12 

12 13 

14 17 

Inferential Statistics 

Dependent t Test  

SAS Code 

 
DATA THROWS; 

  INPUT THROWS_PRE THROWS_POST @@; 

  LINES; 

   12 13 15 15 9 11 

   16 15 12 15 15 18 

   17 17 10 12 12 13 

   14 17 

RUN; 

 

PROC PRINT DATA=THROWS; 

RUN; 

 

(1) PROC MEANS DATA=THROWS; 

RUN; 

 

(2) PROC TTEST DATA=THROWS PLOTS(SHOWH0)=INTERVAL; 

(3)  PAIRED THROWS_PRE*THROWS_POST; 

  TITLE "Dependent t Test: Free Throws"; 

RUN; 

 

TITLE; 

QUIT; 

 

(1) PROC MEANS provides the means and standard deviations for each variable in the 

THROWS dataset (THROWS_PRE and THROWS_POST). 

(2) PROC TTEST will perform a t test on the THROWS data.  PROC TTEST includes 

several useful plots in the output by default.  The SHOWH0 option shows the null 

hypothesis (H0) critical value in the appropriate plots; the INTERVAL option adds an 

additional plot that will aid in interpreting the results.  [Note: PROC TTEST can also be 

used with independent t tests.  The reason that it was omitted from the previous section is 

because PROC TTEST does not offer Levene’s Test to assess the homogeneity of variances 

assumption (it uses a different test).  Therefore, PROC GLM, which defaults to Levene’s 

Test, was used.] 

(3) The PAIRED statement makes this a dependent t test.  The difference scores are 

calculated as the score on the left of the asterisk (*) minus the score on the right of the 

asterisk (*).  In this case, the analysis is THROWS_PRE minus THROWS_POST.  This 

may seem backwards, as the resulting difference values will be negative if the number of 

free throws increases after the treatment.  The reason they have been placed in this order is 

so that THROWS_PRE is placed on the x-axis and THROWS_POST is placed on the y-

axis in applicable output graphs.  The negative sign on the difference values and the 

corresponding t value can be disregarded.  Please note that you can reverse this order if 

you prefer; the only consequence will be the reversal of the variables on the axes of the 

resulting plots. 

Inferential Statistics 

Dependent t Test  

Selected Output 

Inferential Statistics 

One-Way ANOVA 

Research Scenario 

 

A direct marketer of insurance wanted to evaluate the effect of age on the response rate to a new 

insurance product. Below are the response rates per 1,000 mailings by age group from 12 different 

metropolitan areas. 

 

Young Middle-Aged Elderly 

25 30 25 

27 29 22 

23 29 27 

24 31 23 

23 28 24 

24 31 25 

22 29 23 

25 32 22 

21 30 21 

24 29 22 

21 28 24 

23 31 23 

Inferential Statistics 

One-Way ANOVA 

SAS Code 

 
PROC FORMAT; 

(1)  VALUE AGE_FMT 

   1="1. Young" 

   2="2. Middle-Aged" 

   3="3. Elderly"; 

RUN; 

 

DATA INSURANCE; 

  INPUT AGE RATE @@; 

  FORMAT AGE AGE_FMT.; 

  LINES; 

   1 25 1 27 1 23 1 24 1 23 1 24 1 22 1 25 1 21 1 24 1 21 1 23 

   2 30 2 29 2 29 2 31 2 28 2 31 2 29 2 32 2 30 2 29 2 28 2 31 

   3 25 3 22 3 27 3 23 3 24 3 25 3 23 3 22 3 21 3 22 3 24 3 23 

RUN; 

 

PROC PRINT DATA=INSURANCE; 

RUN; 

 

PROC GLM DATA=INSURANCE PLOTS=DIAGNOSTICS; 

(2)  CLASS AGE; 

(3)  MODEL RATE=AGE; 

(4)  MEANS AGE / HOVTEST=LEVENE(TYPE=ABS); 

(5)  LSMEANS AGE / PDIFF=ALL ADJUST=TUKEY; 

  TITLE "One-Way ANOVA: Insurance Response Rate"; 

RUN; 

 

TITLE; 

QUIT; 

 

(1) When labels are applied to the values (code numbers or letters) of a variable, the output 

will show these labels and their corresponding data in alphabetical order.  In this case, 

AGE is coded as 1 = “Young,” 2 = “Middle,” and 3 = “Elderly.”  If these labels were used 

in the VALUE statement (instead of the labels shown in the code above), then the output 

would appear in the order “Elderly,” “Middle,” and “Young” (i.e. alphabetical order).  

This may make the output confusing for the reader, who would expect the axis displaying 

AGE to go from youngest to oldest.  The easiest way to correct this is to include the code 

number in the label (e.g., “1. Young”), as “1” comes first in alphabetical order, followed 

by “2. Middle-Aged” and “3. Elderly.”  Now, alphabetical order and logical order will 

match. 

(2) The CLASS statement identifies AGE as a nominal/categorical/grouping variable. 

(3) The MODEL statement is written as Dependent Variable (DV) = Independent Variable 

(IV). 

(4) The MEANS statement requires SAS to compute the means for each group in AGE.  The 

option HOVTEST=LEVENE calls for the Levene’s Test of Homogeneity of Variances, to 

test the assumption that the groups have equal variances.  There are multiple homogeneity 

tests available; Levene’s Test is the default.  There are two methods for calculating 

Levene’s statistic: using the absolute residuals (TYPE=ABS) or the squared residuals 

(TYPE=SQUARE, the default).  As the users of this guide may be more familiar with 

SPSS or may be referencing SPSS: A Survival Guide for EPRS 8530 and EPRS 8540, the 

TYPE option is set to the method used by SPSS: ABS. 

(5) The LSMEANS statement shown here is only required if the one-way ANOVA is found to 

be significant.  The LSMEANS statement will perform Tukey’s HSD post-hoc testing by 

AGE.  The PDIFF=ALL option requests the p values for each of the pairwise 

comparisons.  If post-hoc testing is not required, you may remove this line of code. 

Inferential Statistics 

One-Way ANOVA 

Selected Output 

Inferential Statistics 

Two-Way ANOVA with Nonsignificant Interaction  

Research Scenario 

 

A marketing manager for a supermarket chain was interested in the effect of both retail price 

and display location on a new promotional line of cookies. A group of 24 stores with matching 

store volume, layout, and customer demographics was split at random into 6 groups of 4 stores 

each. One group was assigned to each of the 6 combinations of retail price (regular retail vs. 

discounted retail) and display location (entrance aisle, cookie aisle, and checkout). Below are 

the average weekly unit sales by store over a 13-week period. 

 

Price 
Display Location 

1. Entrance 2. Cookies 3. Checkout 

1. Regular Retail 

38 28 21 

31 25 32 

27 23 30 

33 20 22 

2. Discounted Retail 

35 22 19 

21 24 15 

39 16 25 

30 17 20 

Inferential Statistics 

Two-Way ANOVA with Nonsignificant Interaction  

SAS Code 

 
PROC FORMAT; 

  VALUE PRICE_FMT 

   1="1. Regular" 

   2="2. Discounted"; 

   VALUE LOC_FMT 

   1="1. Entrance" 

2="2. Cookie Aisle" 

3="3. Checkout"; 

RUN; 

 

DATA NEW_COOKIES; 

(1)  INPUT PRICE LOCATION SALES @@; 

(2)  FORMAT PRICE PRICE_FMT. LOCATION LOC_FMT.; 

  LINES; 

   1 1 38 1 1 31 1 1 27 1 1 33 

   1 2 28 1 2 25 1 2 23 1 2 20 

   1 3 21 1 3 32 1 3 30 1 3 22 

   2 1 35 2 1 21 2 1 39 2 1 30 

   2 2 22 2 2 24 2 2 16 2 2 17 

   2 3 19 2 3 15 2 3 25 2 3 20 

RUN; 

 

PROC PRINT DATA=NEW_COOKIES; 

RUN; 

 

(3) PROC SGPLOT DATA=NEW_COOKIES; 

(4)  VLINE PRICE /  

   GROUP=LOCATION 

   STAT=MEAN 

   RESPONSE=SALES 

   MARKERS; 

RUN; 

 

(5) PROC GLM DATA=NEW_COOKIES ORDER=INTERNAL; 

(6)  CLASS PRICE LOCATION; 

(7)  MODEL SALES=PRICE*LOCATION; 

(8)  MEANS PRICE*LOCATION / HOVTEST=LEVENE(TYPE=ABS); 

  TITLE "Levene's Test of Homogeneity of Variances"; 

RUN; 

 

(9) PROC GLM DATA=NEW_COOKIES ORDER=INTERNAL; 

(10)  CLASS PRICE LOCATION; 

(11)  MODEL SALES=PRICE LOCATION PRICE*LOCATION; 

(12)  MEANS PRICE LOCATION PRICE*LOCATION; 

(13)  LSMEANS LOCATION / PDIFF=ALL ADJUST=TUKEY; 

  TITLE "Two-Way ANOVA: Cookie Sales"; 

RUN; 

 

TITLE; 

QUIT; 

(1) Please take note of the order used for data entry.  PRICE is the variable initiated first, 

then LOCATION, then SALES.  Correspondingly, the raw data are entered with PRICE 

first, LOCATION second, and SALES third.  Therefore, the observations are entered in 

the following order. 

          PRICE = 1 and LOCATION = 1 

          PRICE = 1 and LOCATION = 2 

          PRICE = 1 and LOCATION = 3 

          PRICE = 2 and LOCATION = 1 

          PRICE = 2 and LOCATION = 2 

          PRICE = 2 and LOCATION = 3 

(2) The FORMAT statement is used to apply the labels created in PRICE_FMT and 

LOC_FMT to PRICE and LOCATION, respectively.  Note that a period (.) follows 

PRICE_FMT and LOC_FMT; the code will not work properly if you omit the periods. 

(3) PROC SGPLOT creates a line graph comparing the means by PRICE and LOCATION. 

(4) The VLINE statement specifies PRICE as the variable to be placed on the x-axis.  A 

forward slash (/) is used prior to entering the VLINE options.  The GROUP option 

specifies LOCATION as the variable to use for the lines; each LOCATION will be shown 

as a separate color-coded line.  The STAT option requests the mean for each group to be 

the statistic that is graphed.  The RESPONSE option identifies the y-axis variable, SALES 

in this case.  MARKERS tells SAS to show the means for each group as data points. 

(5) This PROC GLM step is used for Levene’s Test of the homogeneity of variances 

only.  Do NOT use or report any of the other results from this step.  A quirk of SAS is 

that Levene’s Test is only available for one-way ANOVAs.  Therefore, this PROC GLM 

step is written such that the interaction term (PRICE*LOCATION) is the only 

independent variable (IV) in the analysis, making it a one-way ANOVA while retaining 

all six cells whose variances need to be compared.  Once again, do NOT use or report 

any results – other than Levene’s Test – from this step.  Although some of the 

ANOVA results will match the final (correct) results, some of them (including the 

Sum of Squares partitioning) will NOT.  ORDER=INTERNAL tells SAS to analyze the 

data in the order that it was entered in the DATA step. 

(6) The CLASS statement identifies PRICE and LOCATION as the grouping variables. 

(7) The MODEL statement is written as Dependent Variable (DV) = the interaction term 

(PRICE*LOCATION) for the two Independent Variables (IVs).  Remember, this is NOT 

the correct model for a factorial ANOVA.  This MODEL statement is required in order to 

run Levene’s Test. 

(8) The MEANS statement requires SAS to compute the means for each cell of PRICE by 

LOCATION.  The option HOVTEST=LEVENE calls for Levene’s Test of Homogeneity 

of Variances, to test the assumption that the cells have equal variances. 

(9) This PROC GLM step is the correct step for a Factorial ANOVA.  These are the true 

results to use and report for this analysis.  ORDER=INTERNAL tells SAS to analyze 

the data in the order that it was entered in the DATA step. 

(10) The CLASS statement identifies PRICE and LOCATION as the grouping variables. 

(11) This MODEL statement – which is the correct model for a factorial ANOVA – is written 

as Dependent Variable (DV) = Independent Variable 1 (IV1) and Independent Variable 2 

(IV2) and the interaction term for the two IVs (IV1*IV2). 



R 3.4.1: A Survival Guide 

 

57 

 

(12) The MEANS statement requires SAS to compute the means for (A) each group in PRICE, 

(B) each group in LOCATION, and (C) each cell of PRICE by LOCATION. 

(13) The LSMEANS statement shown here is only required if (A) there is a significant 

interaction effect or (B) there is a significant main effect.  Remember, if there is a 

significant interaction effect, then this effect takes priority and you need to conduct post-

hoc analyses to find out which interaction (IV1*IV2) means differ significantly.  If the 

interaction is nonsignificant, you need to examine the main effects for significance; if a 

main effect (or multiple main effects) are significant, conduct post-hoc analyses to find 

out which main effect (IV1 and/or IV2) means differ significantly.  The LSMEANS 

statement will perform Tukey’s HSD post-hoc testing for any effect you specify: 

interaction or main effects.  In this example, the LSMEANS statement performs Tukey’s 

HSD by LOCATION.  The PDIFF=ALL option requests the p values for each of the 

pairwise comparisons. 

Inferential Statistics 

Two-Way ANOVA with Nonsignificant Interaction  

Selected Output 

Inferential Statistics 

Two-Way ANOVA with Significant Interaction  

Research Scenario 

 

Eighteen students were randomly assigned to one of three different classroom teaching methods 

(online, hybrid, or in person) to learn a new math concept.  At the end of 12 weeks, students 

were given a test to assess their understanding of the concept.   A two-way ANOVA will be run 

to assess the difference between the teaching methods as well as between student’s previous 

online experiences.  The table below shows the scores by teaching method and previous 

experience. Two students did not finish the 12-week course so the distribution between the 

teaching methods and the previous experience was not equal. 

 

 1. In person 2. Hybrid 3. Online 

1.  Experienced (Has taken an 

online course in the past) 

57.5 

80.0 

62.5 

72.5 

75.0 

 

77.5 

90.0 

82.5 

2. Not Experienced (Has not 

taken an online class in the past) 

80.0 

65.0 

65.0 

85.0 

100.0 

87.5 

57.5 

65.0 

Inferential Statistics 

Two-Way ANOVA with Significant Interaction  

SAS Code 

 
PROC FORMAT; 

  VALUE EXPER_FMT 

   1="1. Experienced" 

   2="2. Not Experienced"; 

  VALUE METHOD_FMT 

   1="1. In Person" 

   2="2. Hybrid" 

   3="3. Online"; 

RUN; 

 

DATA TEACHING_METHODS; 

(1)  INPUT EXPERIENCE METHOD SCORE @@; 

(2)  FORMAT EXPERIENCE EXPER_FMT. METHOD METHOD_FMT.; 

  LINES; 

   1 1 57.5 1 1 80.0 1 1 62.5 

   1 2 72.5 1 2 75.0 

   1 3 77.5 1 3 90.0 1 3 82.5 

   2 1 80.0 2 1 65.0 2 1 65.0 

   2 2 85.0 2 2 100.0 2 2 87.5 

   2 3 57.5 2 3 65.0 

RUN; 

 

PROC PRINT DATA=TEACHING_METHODS; 

RUN; 

 

(3) PROC SGPLOT DATA=TEACHING_METHODS; 

(4)  VLINE METHOD /  

   GROUP=EXPERIENCE 

   STAT=MEAN 

   RESPONSE=SCORE 

   MARKERS; 

RUN; 

 

(5) PROC GLM DATA=TEACHING_METHODS ORDER=INTERNAL; 

(6)  CLASS EXPERIENCE METHOD; 

(7)  MODEL SCORE=EXPERIENCE*METHOD; 

(8)  MEANS EXPERIENCE*METHOD / HOVTEST=LEVENE(TYPE=ABS); 

  TITLE "Levene's Test of Homogeneity of Variances"; 

RUN; 

 

(9) PROC GLM DATA=TEACHING_METHODS ORDER=INTERNAL; 

(10)  CLASS EXPERIENCE METHOD; 

(11)  MODEL SCORE=EXPERIENCE METHOD EXPERIENCE*METHOD; 

(12)  MEANS EXPERIENCE METHOD EXPERIENCE*METHOD; 

(13)  LSMEANS EXPERIENCE*METHOD / PDIFF=ALL ADJUST=TUKEY; 

  TITLE "Two-Way ANOVA: Math Teaching Methods"; 

RUN; 

 

TITLE; 

QUIT; 

(1) Please take note of the order used for data entry.  EXPERIENCE is the variable initiated 

first, then METHOD, then SCORE.  Correspondingly, the raw data are entered with 

EXPERIENCE first, METHOD second, and SCORE third.  Therefore, the observations 

are entered in the following order. 

          EXPERIENCE = 1 and METHOD = 1 

          EXPERIENCE = 1 and METHOD = 2 

          EXPERIENCE = 1 and METHOD = 3 

          EXPERIENCE = 2 and METHOD = 1 

          EXPERIENCE = 2 and METHOD = 2 

          EXPERIENCE = 2 and METHOD = 3 

(2) The FORMAT statement is used to apply the labels created in EXPER_FMT and 

METHOD _FMT to EXPERIENCE and METHOD, respectively.  Note that a period (.) 

follows EXPER_FMT and METHOD_FMT; the code will not work properly if you omit 

the periods. 

(3) PROC SGPLOT creates a line graph comparing the means by EXPERIENCE and 

METHOD. 

(4) The VLINE statement specifies EXPERIENCE as the variable to be placed on the x-axis.  

A forward slash (/) is used prior to entering the VLINE options.  The GROUP option 

specifies METHOD as the variable to use for the lines; each METHOD will be shown as a 

separate color-coded line.  The STAT option requests the mean for each group to be the 

statistic that is graphed.  The RESPONSE option identifies the y-axis variable, SALES in 

this case.  MARKERS tells SAS to show the means for each group as data points. 

(5) This PROC GLM step is used for Levene’s Test of the homogeneity of variances 

only.  Do NOT use or report any of the other results from this step.  A quirk of SAS is 

that Levene’s Test is only available for one-way ANOVAs.  Therefore, this PROC GLM 

step is written such that the interaction term (EXPERIENCE*METHOD) is the only 

independent variable (IV) in the analysis, making it a one-way ANOVA while retaining 

all six cells whose variances need to be compared.  Once again, do NOT use or report 

any results – other than Levene’s Test – from this step.  Although some of the 

ANOVA results will match the final (correct) results, some of them (including the 

Sum of Squares partitioning) will NOT.  ORDER=INTERNAL tells SAS to analyze the 

data in the order that it was entered in the DATA step. 

(6) The CLASS statement identifies EXPERIENCE and METHOD as the grouping variables. 

(7) The MODEL statement is written as Dependent Variable (DV) = the interaction term 

(EXPERIENCE*METHOD) for the two Independent Variables (IVs).  Remember, this is 

NOT the correct model for a factorial ANOVA.  This MODEL statement is required in 

order to run Levene’s Test. 

(8) The MEANS statement requires SAS to compute the means for each cell of 

EXPERIENCE by METHOD.  The option HOVTEST=LEVENE calls for the Levene’s 

Test of Homogeneity of Variances, to test the assumption that the cells have equal 

variances. 

(9) This PROC GLM step is the correct step for a Factorial ANOVA.  These are the 

correct results to use and report for this analysis.  ORDER=INTERNAL tells SAS to 

analyze the data in the order that it was entered in the DATA step. 

(10) The CLASS statement identifies EXPERIENCE and METHOD as the grouping variables. 

(11) This MODEL statement – which is the correct model for a factorial ANOVA – is written 

as Dependent Variable (DV) = Independent Variable 1 (IV1) and Independent Variable 2 

(IV2) and the interaction term for the two IVs (IV1*IV2). 

(12) The MEANS statement requires SAS to compute the means for (A) each group in 

EXPERIENCE, (B) each group in METHOD, and (C) each cell of EXPERIENCE by 

METHOD. 

(13) The LSMEANS statement shown here is only required if (A) there is a significant 

interaction effect or (B) there is a significant main effect.  Remember, if there is a 

significant interaction effect, then this effect takes priority and you need to conduct post-

hoc analyses to find out which interaction (IV1*IV2) means differ significantly.  If the 

interaction is nonsignificant, you need to examine the main effects for significance; if a 

main effect (or multiple main effects) are significant, conduct post-hoc analyses to find 

out which main effect (IV1 and/or IV2) means differ significantly.  The LSMEANS 

statement will perform Tukey’s HSD post-hoc testing for any effect you specify: 

interaction or main effects.  In this example, the LSMEANS statement performs Tukey’s 

HSD by METHOD.  The PDIFF=ALL option requests the p values for each of the 

pairwise comparisons.  

Inferential Statistics 

Two-Way ANOVA with Significant Interaction  

Selected Output 

Inferential Statistics 

Analysis of Covariance (ANCOVA) 

Research Scenario 

 

A personnel manager wished to evaluate the effect of positive and negative reinforcement on 

tardiness in a large manufacturing plant. A group of 30 chronic late arrivers were identified 

based on their previous tardiness records. They were then randomly assigned in equal numbers 

to one of three study groups: (1) positive reinforcement when on time, (2) negative 

reinforcement when late, and (3) no reinforcement. After a 10-week treatment period, data was 

collected on the number of tardies over an additional 10-week period. The data below presents 

the number of tardies for 10 weeks prior to the intervention and 10 weeks following termination 

of the treatment. 

 

Type of Reinforcement 

1. Positive Reinforcement 2. Negative Reinforcement 3. No Reinforcement 

Pre Post Pre Post Pre Post 

4 2 4 3 5 5 

5 3 5 5 5 4 

6 1 5 3 5 4 

7 4 6 6 6 8 

7 3 7 6 8 8 

8 5 8 7 8 9 

9 3 8 6 9 7 

9 5 9 8 10 10 

10 6 10 6 10 8 

11 5 11 4 7 10 

Inferential Statistics 

Analysis of Covariance (ANCOVA) 

SAS Code 

 
PROC FORMAT; 

  VALUE REINF_FMT 

   1="1. Positive" 

   2="2. Negative" 

   3="3. None"; 

RUN; 

 

DATA TARDINESS_REINF; 

  INPUT REINF PRE POST @@; 

  FORMAT REINF REINF_FMT.; 

  LINES; 

   1 4 2 1 5 3  

1 6 1 1 7 4  

1 7 3 1 8 5  

1 9 3 1 9 5  

1 10 6 1 11 5 

   2 4 3 2 5 5  

2 5 3 2 6 6  

2 7 6 2 8 7  

2 8 6 2 9 8  

2 10 6 2 11 4 

   3 5 5 3 5 4  

3 5 4 3 6 8  

3 8 8 3 8 9  

3 9 7 3 10 10  

3 10 8 3 7 10 

RUN; 

 

PROC PRINT DATA=TARDINESS_REINF; 

RUN; 

 
(1) PROC GLM DATA=TARDINESS_REINF PLOTS=DIAGNOSTICS; 

  CLASS REINF; 

  MODEL POST=REINF; 

  MEANS REINF / HOVTEST=LEVENE(TYPE=ABS); 

  TITLE "Levene's Test of Homogeneity of Variances"; 

RUN; 

 

(2) PROC GLM DATA=TARDINESS_REINF; 

  CLASS REINF; 

  MODEL POST=REINF PRE REINF*PRE; 

  TITLE "Test of Assumption of Homogeneity of Slopes"; 

RUN; 

 

(3) PROC GLM DATA=TARDINESS_REINF; 

  CLASS REINF; 

  MODEL POST=REINF PRE; 

  TITLE "ANCOVA: Employee Tardiness"; 

RUN; 

(4) PROC GLM DATA=TARDINESS_REINF; 

  CLASS REINF; 

  MODEL POST=REINF PRE; 

  LSMEANS REINF / PDIFF ADJUST=TUKEY; 

  TITLE "Adjusted Means & Tukey Post-Hoc Comparisons"; 

RUN; 

 

(5) PROC GLM DATA=TARDINESS_REINF; 

  CLASS REINF; 

  MODEL PRE=REINF; 

  TITLE "Bryant-Paulson (BP) F Value"; 

RUN; 

 

TITLE; 

QUIT; 

 

(1) This PROC GLM step is used for Levene’s Test of the homogeneity of variances.  Do 

NOT use or report any of the other significance test results from this step.  It will 

perform Levene’s Test by conducting a one-way ANOVA of POST on REINF.  It will also 

produce graphics to help determine whether the assumption of normality is tenable.  

Additionally, the unadjusted means and standard deviations will be produced by this 

procedure. 

(2) This PROC GLM step is used to test the homogeneity of slopes assumption.  Do not use 

any other results from this procedure.  An ANCOVA test is produced by this PROC 

GLM by including an interaction term (REINF*PRE) in the model. 

(3) This PROC GLM step is the correct step for ANCOVA.  These are the correct results 

to use and report for this analysis.  This PROC GLM is the same as that in (2), except that 

the interaction term is excluded.  The output from this PROC GLM should be reported as 

the ANCOVA results. 

(4) This PROC GLM step is used to generate the adjusted pretest/pre-treatment means 

and Tukey’s post-hoc comparisons.  The LSMEANS (“least squares” or “estimated 

marginal” means) statement requests the adjusted means, as well as the post-hoc 

comparisons with a Tukey alpha level adjustment for multiple significance tests.   

(5) This PROC GLM step is used to calculate the F value for the Bryant-Paulson (BP) 

Post-Hoc Procedure.  Do not use any other results from this procedure.  If you want to 

conduct the BP Procedure, you can use PROC GLM to run a one-way ANOVA modeling 

PRE as the dependent variable and REINF as the independent variable.  This will provide 

the F value that you will need for the manual BP calculations. 

Inferential Statistics 

Analysis of Covariance (ANCOVA) 

Selected Output 

Inferential Statistics 

Repeated Measures: One Within Factor Design  

Research Scenario 

 

A high school math teacher studies the impact of paper color on mathematics test scores. The 

hypothesis was test scores would be higher on tests taken on pastel green paper than tests taken 

on bright yellow paper or traditional white paper because the cool color would have a calming 

effect and reduce test-taking anxiety. Weekly math tests for an Algebra I class were printed in 

equal quantities on the three colors of paper. The order of treatment was counterbalanced such 

that one third of the students were randomly assigned to a different color each week over a 

three-week period. Below are the test scores by student by paper color. 

 

Student 
Paper Color 

Yellow (Y) Green (G) White (W) 

1 80 76 77 

2 81 89 70 

3 39 64 55 

4 95 93 91 

5 71 90 87 

6 86 76 92 

7 98 94 83 

8 95 92 92 

9 73 77 53 

10 78 88 83 

11 54 64 57 

12 73 92 96 

13 82 75 69 

14 49 67 55 

15 83 91 79 

16 91 90 88 

17 94 97 91 

18 85 89 90 

19 62 69 45 

Inferential Statistics 

Repeated Measures: One Within Factor Design  

SAS Code 

 
DATA PAPER_COLOR; 

  INPUT ID YELLOW GREEN WHITE; 

  LINES; 

   1 80 76 77 

   2 81 89 70 

   3 39 64 55 

   4 95 93 91 

   5 71 90 87 

   6 86 76 92 

   7 98 94 83 

   8 95 92 92 

   9 73 77 53 

   10 78 88 83 

   11 54 64 57 

   12 73 92 96 

   13 82 75 69 

   14 49 67 55 

   15 83 91 79 

   16 91 90 88 

   17 94 97 91 

   18 85 89 90 

   19 62 69 45 

RUN; 

 

PROC PRINT DATA=PAPER_COLOR; 

RUN; 

 

PROC MEANS DATA=PAPER_COLOR; 

RUN; 

 

(1) PROC GLM DATA=PAPER_COLOR ORDER=INTERNAL; 

(2)  MODEL YELLOW GREEN WHITE= / NOUNI; 

(3)  REPEATED COLOR / PRINTE; 

  TITLE "Repeated Measures ANOVA: Paper Color"; 

RUN; 

 

 TITLE; 

 

PROC FORMAT; 

  VALUE COLOR_FMT 

   1="Yellow" 

   2="Green" 

   3="White"; 

RUN; 

 

(4) DATA PAPER_COLOR_2; 

(5)  SET PAPER_COLOR; 

(6)  SCORE=YELLOW; COLOR=1; OUTPUT; 

(7)  SCORE=GREEN; COLOR=2; OUTPUT; 

(8)  SCORE=WHITE; COLOR=3; OUTPUT; 

(9)  FORMAT COLOR COLOR_FMT.; 

(10)  DROP YELLOW GREEN WHITE; 

RUN; 

 

(11) PROC PRINT DATA=PAPER_COLOR_2; 

RUN; 

 

(12) PROC SGPANEL DATA=PAPER_COLOR_2; 

(13)  PANELBY COLOR / UNISCALE=ROW; 

(14)  HISTOGRAM SCORE; 

(15)  DENSITY SCORE; 

RUN; 

 

(16) PROC GLM DATA=PAPER_COLOR_2 ORDER=INTERNAL PLOTS=DIAGNOSTICS; 

  CLASS ID COLOR; 

(17)  MODEL SCORE=ID COLOR; 

(18)  LSMEANS COLOR / PDIFF CL ADJUST=BON; 

  TITLE "Bonferroni Post-Hoc Comparisons"; 

RUN; 

 

TITLE; 

QUIT; 

 

(1) This PROC GLM step is the correct step for repeated measures ANOVA.  These are 

the correct results to use and report for this analysis.  The output from this PROC 

GLM should be reported as the repeated measures ANOVA results. 

(2) As always, the MODEL statement is written as DV(s) = IV(s).  Repeated measures 

ANOVA can be approached from a univariate (single DV) perspective or a multivariate 

(multiple DV) perspective.  It is for this reason that you will see the repeated measures 

variable(s) placed in the MODEL statement to the left of the equal sign (=) where the DVs 

belong.  In this case, the colors (YELLOW, GREEN, and WHITE) are the within 

(repeated) factors, so they are treated as the DVs.  There are no between (grouping) 

factors, so there are no IVs; there is nothing to the right of the equal sign.  The NOUNI 

option suppresses some univariate output that you will not need. 

(3) The REPEATED statement is what makes this a repeated measures analysis.  For the 

purpose of making the output more informative, you can follow the statement 

REPEATED with a word to describe/name the repeated measure being analyzed; in this 

case, it was named COLOR.  (TIME, TREATMENT, and TRIAL may be good options in 

other circumstances.)  It is important for you to understand that this word is not a variable; 

it is just a label that SAS will use in the output.  The PRINTE option produces 

supplemental output, including Mauchly’s Test of Sphericity. 

(4) Unfortunately, if you need Bonferroni post-hoc testing of the within factor(s), you cannot 

get it with the previous PROC GLM code.  In order to get Bonferroni results, you will first 

need to “reshape” your data; you will need to go from the “wide” format you began with 

to a “long” format.  In this DATA step, a long data format is created and named 

PAPER_COLOR_2. 

(5) The SET statement copies the data from PAPER_COLOR into PAPER_COLOR_2. 

(6) Two new variables are created for the PAPER_COLOR_2 dataset: SCORE and COLOR.  

SAS copies each YELLOW value to SCORE and simultaneously codes that observation 

as a “1” for COLOR. 

(7) SAS copies each GREEN value to SCORE and simultaneously codes that observation as a 

“2” for COLOR. 

(8) SAS copies each WHITE value to SCORE and simultaneously codes that observation as a 

“3” for COLOR. 

(9) The FORMAT statement is used to apply the labels created in COLOR_FMT to COLOR.  

Note that a period (.) follows COLOR_FMT; the code will not work properly if you omit 

the period. 

(10) The DROP statement deletes the variables YELLOW, GREEN, and WHITE from 

PAPER_COLOR_2 because they are no longer needed (because COLOR was created and 

coded as 1, 2, or 3).  Note: These variables still exist in the original dataset 

PAPER_COLOR. 

(11) Again, it is always a good idea to display your new dataset and confirm there are no data 

entry errors. 

(12) PROC SGPANEL creates a panel of graphs.  This will be used to assess whether SCORE 

is normally distributed for each COLOR. 

(13) The PANELBY statement requests that each panel represent one level of COLOR.  The 

UNISCALE=ROW option requests that all panels have the same x-axis scale. 

(14) The HISTOGRAM statement will produce histograms of SCORE.  These histograms will 

be produced by COLOR per the statement in (13). 

(15) The DENSITY statement will overlay a normal curve on each histogram. 

(16) This PROC GLM step is used generate the Bonferroni post-hoc comparisons.  Do 

NOT use or report any of the other results from this step.  Some of the results from 

this PROC GLM will match the repeated measures ANOVA results, but some of the 

results are different.  Only the Bonferroni post-hoc comparisons should be reported. 

(17) The MODEL statement includes the ID variable as an IV.  Note that the new variable 

SCORE is the DV and the new grouping variable COLOR is an IV. 

(18) The LSMEANS (“least squares” or “estimated marginal” means) statement requests the 

adjusted means, as well as the Bonferroni post-hoc comparisons.  The CL option requests 

confidence limits (confidence intervals) for the results.  

Inferential Statistics 

Repeated Measures: One Within Factor Design  

Selected Output 

Inferential Statistics 

Repeated Measures: One Within Factor and One Between Factor Design 

Research Scenario 

 

A researcher wanted to investigate the effect of anxiety on math performance for fourth graders 

in a variety of testing time constraints.  She came up with a 20-question multiplication test and 

gave it to twelve participants.  For the first trial, she allowed participants one hour.  For the 

second trial, participants had 45 minutes to take the test.  The time for the third trial was 30 

minutes.  The time for the final trial was 15 minutes.  Prior to giving the math test, she assessed 

students’ test anxiety level; students with low test anxiety were in Group 1 and those with high 

test anxiety were in Group 2. The tests, however, were the same between groups. The data can 

be seen below. 

 

Subject Anxiety Trial 1 Trial 2 Trial 3 Trial 4 

1 1 18 14 12 6 

2 1 19 12 8 4 

3 1 14 10 6 2 

4 1 16 12 10 4 

5 1 12 8 6 2 

6 1 18 10 5 1 

7 2 16 10 8 4 

8 2 18 8 4 1 

9 2 16 12 6 2 

10 2 19 16 10 8 

11 2 16 14 10 9 

12 2 16 12 8 8 

Inferential Statistics 

Repeated Measures: One Within Factor and One Between Factor Design 

SAS Code 

 
PROC FORMAT; 

  VALUE ANX_FMT 

   1="1. Low Anxiety" 

   2="2. High Anxiety"; 

RUN; 

 

DATA TESTING_TIME; 

  INPUT ID ANXIETY TRIAL1 TRIAL2 TRIAL3 TRIAL4; 

  FORMAT ANXIETY ANX_FMT.; 

  LINES; 

   1 1 18 14 12 6 

   2 1 19 12 8 4 

   3 1 14 10 6 2 

   4 1 16 12 10 4 

   5 1 12 8 6 2 

   6 1 18 10 5 1 

   7 2 16 10 8 4 

   8 2 18 8 4 1 

   9 2 16 12 6 2 

   10 2 19 16 10 8 

   11 2 16 14 10 9 

   12 2 16 12 8 8 

RUN; 

 

PROC PRINT DATA=TESTING_TIME; 

RUN; 

 

PROC MEANS DATA=TESTING_TIME; 

(1)  BY ANXIETY; 

RUN; 

 

(2) PROC GLM DATA=TESTING_TIME; 

  CLASS ANXIETY; 

(3)  MODEL TRIAL1-TRIAL4=ANXIETY / NOUNI; 

(4)  MEANS ANXIETY / HOVTEST=LEVENE(TYPE=ABS); 

(5)  REPEATED TRIAL / PRINTE; 

(6)  LSMEANS ANXIETY / PDIFF CL ADJUST=BON; 

  TITLE "Repeated Measures ANOVA: Testing Time"; 

RUN; 

 

TITLE; 

 

(7) DATA TESTING_TIME_2; 

  SET TESTING_TIME; 

  SCORE=TRIAL1; TRIAL=1; OUTPUT; 

  SCORE=TRIAL2; TRIAL=2; OUTPUT; 

  SCORE=TRIAL3; TRIAL=3; OUTPUT; 

  SCORE=TRIAL4; TRIAL=4; OUTPUT; 

  DROP TRIAL1-TRIAL4; 

RUN; 

PROC PRINT DATA=TESTING_TIME_2; 

RUN; 

 

(8) PROC GLM DATA=TESTING_TIME_2 ORDER=INTERNAL PLOTS=DIAGNOSTICS; 

  CLASS ID ANXIETY TRIAL; 

(9)  MODEL SCORE=ID ANXIETY TRIAL; 

(10)  LSMEANS TRIAL / PDIFF CL ADJUST=BON; 

  TITLE "Bonferroni Post-Hoc Comparisons"; 

RUN; 

 

TITLE; 

QUIT; 

 

(1) The BY statement requires PROC MEANS to calculate statistics for each level of 

ANXIETY; in other words, it will calculate statistics for each TRIAL*ANXIETY cell. 

(2) This PROC GLM step is the correct step for repeated measures ANOVA.  These are 

the correct results to use and report for this analysis.  The output from this PROC 

GLM should be reported as the repeated measures ANOVA results. 

(3) As always, the MODEL statement is written as DV(s) = IV(s).  Repeated measures 

ANOVA can be approached from a univariate (single DV) perspective or a multivariate 

(multiple DV) perspective.  It is for this reason that you will see the repeated measures 

variable(s) placed in the MODEL statement to the left of the equal sign (=) where the DVs 

belong.  In this case, the trials (TRIAL1, TRIAL2, TRIAL3, and TRIAL4) are the within 

(repeated) factors, so they are treated as the DVs.  You have the option of writing each of 

these DVs individually, but writing TRIAL1-TRIAL4 is quicker.  ANXIETY is a between 

(grouping) factor, which is treated as an IV and written to the right of the equal sign.  The 

NOUNI option suppresses some univariate output that you will not need. 

(4) The MEANS statement with the option HOVTEST=LEVENE will conduct Levene’s Test 

of Homogeneity of Variances, to test the assumption that the cells have equal variances, 

for TRIAL1-TRIAL4. 

(5) The REPEATED statement is what makes this a repeated measures analysis.  For the 

purpose of making the output more informative, you can follow the statement 

REPEATED with a word to describe/name the repeated measure being analyzed; in this 

case, it was named TRIAL.  (TIME and TREATMENT may be good options in other 

circumstances.)  It is important for you to understand that this word is not a variable; it is 

just a label that SAS will use in the output.  The PRINTE option produces supplemental 

output, including Mauchly’s Test of Sphericity. 

(6) The LSMEANS (“least squares” or “estimated marginal” means) statement requests the 

adjusted means, as well as the Bonferroni post-hoc comparisons, for the between factor(s).  

The CL option requests confidence limits (confidence intervals) for the results. 

(7) Unfortunately, if you need Bonferroni post-hoc testing of the within factor(s), you cannot 

get it with the previous PROC GLM code.  In order to get Bonferroni results, you will first 

need to “reshape” your data; you will need to go from the “wide” format you began with 

to a “long” format.  In this DATA step, a long data format is created and named 

TESTING_TIME_2. 

(8) This PROC GLM step is used generate the Bonferroni post-hoc comparisons.  Do 

NOT use or report any of the other results from this step.  Some of the results from 

this PROC GLM will match the repeated measures ANOVA results, but some of the 

results are different.  Only the Bonferroni post-hoc comparisons should be reported. 

(9) The MODEL statement includes the ID variable as an IV.  Note that the new variable 

SCORE is the DV and the new grouping variable TRIAL is an IV. 

(10) The LSMEANS (“least squares” or “estimated marginal” means) statement requests the 

adjusted means, as well as the Bonferroni post-hoc comparisons.  The CL option requests 

confidence limits (confidence intervals) for the results.  

Inferential Statistics 

Repeated Measures: One Within Factor and One Between Factor Design 

Selected Output 

Inferential Statistics 

Simple Linear Regression  

Research Scenario 

 

A cereal-lover would like to lose weight and is interested to know if the amount of complex 

carbohydrates in his cereal significantly affects the cereal’s calorie count.  The complex 

carbohydrates are measured in grams; carbohydrates and calories values are per serving. 

 

This analysis will use the “UScereal.csv” file that can be found as a companion with this SAS 

guide.  If you prefer to enter the data manually, instead of reading in the data file, the raw data 

for the variables of interest may be found in the Appendix of this guide. 

 

 
 

Source of Data: StatLib. (1993). Serial correlation or cereal correlation?? [Data file]. 

Retrieved from http://lib.stat.cmu.edu/datasets/1993.expo/  

Inferential Statistics 

Simple Linear Regression  

SAS Code 

 
DATA US_CEREAL; 

(1)  INFILE "C:\UScereal.CSV" FIRSTOBS=2 DSD MISSOVER; 

(2)  INPUT CEREAL $ MFR $ CALORIES PROTEIN FAT SODIUM  

   FIBER CARBO SUGARS SHELF POTASSIUM VITAMINS $; 

(3)  KEEP CALORIES CARBO; 

RUN; 

 

PROC PRINT DATA=US_CEREAL; 

RUN; 

 

PROC CORR DATA=US_CEREAL PLOTS=SCATTER(ELLIPSE=NONE); 

RUN; 

 

(4) PROC GLM DATA=US_CEREAL PLOTS=(DIAGNOSTICS RESIDUALS); 

(5)  MODEL CALORIES=CARBO / CLPARM; 

  TITLE "Simple Linear Regression: US Cereal"; 

RUN; 

 

(6) PROC REG DATA=US_CEREAL  

(7)   PLOTS(LABEL)=(COOKSD RSTUDENTBYLEVERAGE DFFITS DFBETAS); 

(8)  MODEL CALORIES=CARBO; 

  TITLE "Simple Linear Regression: US Cereal"; 

RUN; 

 

TITLE; 

QUIT; 

 

Note: There are two primary ways of performing a simple linear regression in SAS: 

PROC GLM and PROC REG.  Both methods return the exact same results, so you can 

use either procedure.  However, there are some supplemental results provided by PROC 

REG that are not provided by PROC GLM (e.g., adjusted R-square) and vice versa, so 

you may want to run both simultaneously so that you have everything you need.  Both 

procedures are modeled here. 

 

(1) The INFILE statement is used to identify the file path and file name of the raw data file.  

The optional statement FIRSTOBS=2 tells SAS that the first observation is located on 

Row 2 (in this case, because Row 1 of the data file contains the variable names).  The 

DSD (delimiter-separated data) option performs two functions here.  First, it tells SAS that 

the data is separated by commas (i.e. it is a .CSV file).  Second, if two consecutive 

commas are found, it forces SAS to treat that as missing data.  If the end of an observation 

(row) is blank, there will not be two consecutive commas, but you still need to treat it as 

missing data; the MISSOVER option is used to accomplish this.  

(2) In the same manner as when you manually entered data, the INPUT statement creates the 

variable names and assigns the order of the variables to the new dataset.  This command 

also assigns a variable type to each new variable.  The default variable type is numeric.  If 

a character variable is being created, put a dollar sign ($) in back of it. 

(3) You must read in every variable in order.  However, for the purpose of this analysis, we 

only need to KEEP two of the variables: CALORIES and CARBO. 

(4) This is the PROC GLM method of simple linear regression. 

(5) The MODEL statement is written as Dependent Variable (DV) = Independent Variable 

(IV).  The CLPARM option requests the confidence limits (confidence interval) for the 

parameter estimates. 

(6) This is the PROC REG method of simple linear regression. 

(7) Both PROC REG and PROC GLM include diagnostic plots that help determine if outliers 

are exerting undue influence on the regression analysis.  However, only PROC REG offers 

the LABEL option.  This option requests that the observation numbers of influential 

observations be included in the graphics, making it much easier to determine which 

observations (if any) are impacting the results. 

(8) The MODEL statement is written as Dependent Variable (DV) = Independent Variable 

(IV).  

Inferential Statistics 

Simple Linear Regression  

Selected Output 

Inferential Statistics 

Multiple Regression  

Research Scenario 

 

A botanist decides to test whether temperature and rainfall are good predictors of plant height. 

 

This analysis will use the “Plant_Height.csv” file that can be found (A) as a companion with 

this SAS guide or (B) at the website below by scrolling down to the “Running the analysis” 

section and clicking on the download link.  Please be aware that, if you choose to download the 

file from the website, you will need to complete the following “cleaning” of the dataset. 

1. Remove all commas from Column I (“Site”).  You will be reading this file as a .CSV 

(comma-separated value) file.  These commas in the data would cause SAS to act as 

though it was reading data for the variable in Column J when it was still reading Column 

I. 

2. Change all “NA” values to blanks or periods in Columns AG and AH.  This is because 

Columns AG and AH are numeric data.  In SAS, a period is used to represent missing 

numeric data.  (“NA” would be considered character data.) 

3. Column B is named “site” (lowercase) and Column I is named “Site” (with a capital).  

You do not have to change anything, but be aware that the SAS code herein is written 

such that the variable in Column I will be called SITE_NAME to differentiate the two 

variables. 

If you use the “Plant_Height.csv” file that can be found as a companion with this SAS guide, 

this cleaning has already been completed for you. 

 

If you prefer to enter the data manually, instead of reading in the data file, the raw data for the 

variables of interest may be found in the Appendix of this guide. 

 

 
 

Source of Data: Letten, A. (2016). Linear regression. Retrieved from 

http://environmentalcomputing.net/linear-regression/  

Inferential Statistics 

Multiple Regression  

SAS Code 

 
DATA PLANT_HEIGHT; 

(1)  INFILE "C:\Plant_Height.CSV" FIRSTOBS=2 DSD MISSOVER; 

(2)  INPUT SORT_NUMBER SITE GENUS_SPECIES $ FAMILY $ GROWTHFORM $  

   HEIGHT LOGHT COUNTRY $ SITE_NAME $ LAT LONG ENTERED_BY $ 

   ALT TEMP DIURN_TEMP ISOTHERM TEMP_SEAS TEMP_MAX_WARM  

   TEMP_MIN_COLD TEMP_ANN_RANGE TEMP_MEAN_WETQR  

   TEMP_MEAN_DRYQR TEMP_MEAN_WARMQR TEMP_MEAN_COLDQR RAIN  

   RAIN_WETM RAIN_DRYM RAIN_SEAS RAIN_WETQR RAIN_DRYQR  

   RAIN_WARMQR RAIN_COLDQR LAI NPP HEMISPHERE; 

(3)  KEEP HEIGHT LOGHT TEMP RAIN; 

RUN; 

 

PROC PRINT DATA=PLANT_HEIGHT; 

RUN; 

 

PROC CORR DATA=PLANT_HEIGHT PLOTS=SCATTER(ELLIPSE=NONE); 

RUN; 

 

(4) DATA PLANT_HEIGHT; 

(5)  SET PLANT_HEIGHT; 

(6)  DROP HEIGHT; 

RUN; 

 

PROC PRINT DATA=PLANT_HEIGHT; 

RUN; 

 

PROC CORR DATA=PLANT_HEIGHT PLOTS=SCATTER(ELLIPSE=NONE); 

RUN; 

 

(7) PROC GLM DATA=PLANT_HEIGHT PLOTS=(DIAGNOSTICS RESIDUALS); 

(8)  MODEL LOGHT=TEMP RAIN / CLPARM; 

(9)  OUTPUT OUT=INFLUENCE_RESULTS  

(10)    PREDICTED=Y_PRED  

    RESIDUAL=RESID  

    STUDENT=SRESID  

    RSTUDENT=SDRESID  

    H=HAT_H  

    COOKD=COOKS_D  

    DFFITS=DFFITS  

    LCLM=LCL_MEAN  

    UCLM=UCL_MEAN  

    LCL=LCL_INDIVID  

    UCL=UCL_INDIVID  

    PRESS=PRESS; 

  TITLE "Multiple Regression: Plant Height"; 

RUN; 

 

TITLE; 

 

(11) PROC PRINT DATA=INFLUENCE_RESULTS; 

RUN; 

 

(12) PROC CORR DATA=INFLUENCE_RESULTS; 

(13)  VAR TEMP RAIN RESID; 

RUN; 

 

(14) PROC EXPORT DATA=INFLUENCE_RESULTS 

(15)  OUTFILE="C:\temp\PlantHeight_Influence.CSV" 

(16)  REPLACE; 

RUN; 

 

(17) PROC REG DATA=PLANT_HEIGHT  

(18)   PLOTS(LABEL)=(COOKSD RSTUDENTBYLEVERAGE DFFITS DFBETAS); 

(19)  MODEL LOGHT=TEMP RAIN / INFLUENCE TOL VIF PARTIAL; 

  TITLE "Multiple Regression: Plant Height"; 

RUN; 

 

TITLE; 

 

(20) DATA PLANT_HEIGHT_2; 

(21)  SET PLANT_HEIGHT; 

(22)  ID = _N_; 

(23)  IF ID=22 OR ID=102 OR ID=146 THEN DELETE; 

RUN; 

 

(24) PROC PRINT DATA=PLANT_HEIGHT_2; 

RUN; 

 
TITLE; 

QUIT; 
 

Note 1: There are two primary ways of performing a simple linear regression in SAS: 

PROC GLM and PROC REG.  Both methods return the exact same results, so you can 

use either procedure.  However, there are some supplemental results provided by PROC 

REG that are not provided by PROC GLM (e.g., adjusted R-square) and vice versa, so 

you may just want to run both simultaneously so that you have everything you need.  Both 

procedures are modeled here. 

 

Note 2: The code for multiple regression is EXACTLY the same as the code for simple 

linear regression except that there is at least one more IV in the MODEL statement [Lines 

(8) and (19) of this example].  The multiple regression code presented here is 

comprehensive.  It includes additional (optional) code for conducting diagnostic analyses, 

predictive analyses, etc., topics which you will learn throughout EPRS 8550.  If you are in 

the first few weeks of the course and are learning multiple regression for the first time, 

you are welcome to use the simple linear regression code from the previous section of this 

SAS guide.  All you need to do is make sure you include all of your IVs in the PROC GLM 

and PROC REG MODEL statements.  😊 

 

(1) The INFILE statement is used to identify the file path and file name of the raw data file.  

The optional statement FIRSTOBS=2 tells SAS that the first observation is located on 

Row 2 (in this case, because Row 1 of the data file contains the variable names).  The 

DSD (delimiter-separated data) option performs two functions here.  First, it tells SAS that 

the data is separated by commas (i.e. it is a .CSV file).  Second, if two consecutive 

commas are found, it forces SAS to treat that as missing data.  If the end of an observation 

(row) is blank, there will not be two consecutive commas, but you still need to treat it as 

missing data; the MISSOVER option is used to accomplish this.  

(2) In the same manner as when you manually entered data, the INPUT statement creates the 

variable names and assigns the order of the variables to the new dataset.  This command 

also assigns a variable type to each new variable.  The default variable type is numeric.  If 

a character variable is being created, put a dollar sign ($) in back of it. 

(3) You must read in every variable in order.  However, for the purpose of this analysis, we 

only need to KEEP four of the variables: HEIGHT, LOGHT, TEMP, and RAIN. 

(4) From the output shown below, you will see that it is prudent to drop HEIGHT from the 

model.  This DATA step is used to rewrite the PLANT_HEIGHT dataset without 

HEIGHT. 

(5) The data from the previous PLANT_HEIGHT dataset is copied to the new 

PLANT_HEIGHT dataset using the SET statement. 

(6) The DROP statement is used to drop HEIGHT from the dataset. 

(7) This is the PROC GLM method of simple linear regression. 

(8) The MODEL statement is written as Dependent Variable (DV) = Independent Variable 

(IV).  The CLPARM option requests the confidence limits (confidence interval) for the 

parameter estimates. 

(9) PROC GLM offers a wide variety of influence statistics, some of which are not available 

in PROC REG.  However, it will not produce influence statistics in the output window.  

Instead, you must request that SAS create an output dataset by using the OUTPUT 

statement.  Next, use OUT= to assign that dataset a name; in this case, the name of the 

new dataset is INFLUENCE_RESULTS. 

(10) Beginning with Line (10), you need to identify every influence statistic you want included 

in the new dataset; this will be written to the left of the equal sign.  Then, you need to 

assign each statistic the name you want it to have in your new dataset; this will be written 

to the right of the equal sign.  For example, in Line (10), the PREDICTED statistic is 

being requested; it will be called Y_PRED in the new dataset.  (Note: You can use the 

same name on the left and right of the equal sign if you want.)  All of the influence 

statistics included in this code will be taught in EPRS 8550.  😊 

• PREDICTED: The predicted value of Y (Y’) 

• RESIDUAL: The residual (Y – Y’) 

• STUDENT: The residual divided by its standard error (studentized residuals; SRESID) 

• RSTUDENT: A studentized residual with the current observation deleted (SDRESID) 

• H: Leverage 

• COOKD: Cook’s D 

• DFFITS: Standard influence of observation on predicted value 

• LCLM: Lower confidence limit for the conditional mean 

• UCLM: Upper confidence limit for the conditional mean 

• LCL: Lower confidence limit for an individual prediction 

• UCL: Upper confidence limit for an individual prediction 

• PRESS: Residual for the ith observation that results from dropping it and predicting it 

on the basis of all other observations. This is the residual divided by (1 – hi), where hi 

is the leverage. 

(11) PROC PRINT is used to display the new INFLUENCE_RESULTS dataset in the output 

window. 

(12) One assumption of multiple regression is that the IVs are independent of (i.e. not 

correlated to) the model residuals.  PROC CORR will generate the correlations to test this 

assumption. 

(13) The VAR statement tells SAS to generate the correlations for TEMP, RAIN, and RESID 

(the model residuals from the INFLUENCE_RESULTS dataset). 

(14) In order to find the most influential observations in a large dataset, you may need to sort 

your influence diagnostics.  For example, you may want to sort the studentized residuals 

from greatest to least.  You cannot do this in the output window.  You may copy and paste 

the data from the output window into Excel.  Another option is to export the data.  PROC 

EXPORT creates a data file that you can then open and manipulate in Excel.  In this case, 

you want to export the INFLUENCE_RESULTS data. 

(15) The OUTFILE statement assigns a file path, file name, and file extension. 

(16) The REPLACE statement is optional.  It tells SAS to replace (overwrite) an existing file of 

the same name in the same location (if there is one).  It is recommended that you include 

this option in case you make changes to your data or your analysis, because SAS will NOT 

give you an error or a warning if there is an existing file and you do not replace it.  (The 

log will report this as a blue note, which blends in with all of the other successful log 

notes.  You may think that you overwrote the file, when you did not.) 

(17) This is the PROC REG method of simple linear regression. 

(18) Both PROC REG and PROC GLM include diagnostic plots that help determine if outliers 

are exerting undue influence on the regression analysis.  However, only PROC REG offers 

the LABEL option.  This option requests that the observation numbers of influential 

observations be included in the graphics, making it much easier to determine which 

observations (if any) are impacting the results. Notice also that PROC REG has a 

DFBETA plot.  PROC REG offers DFBETA plots and statistics; PROC GLM does not. 

(19) Obtaining influence statistics in PROC REG is much simpler than in PROC GLM; all you 

have to do is add the INFLUENCE option to the MODEL statement.  The influence 

statistics will appear in the output window.  If you want to sort them, you can copy and 

paste them into Excel.  The TOL option requests tolerance statistics; the VIF option 

requests VIF statistics.  The PARTIAL option is used to generate partial regression plots 

(also called “added variable plots”). 

(20) As you can see from the output screenshots in the next section of this guide, there are 

several data points that may be considered influential.  Suppose you decide that it is 

appropriate to remove Observations #22, 102, and 146 from the analysis.  In order to 

analyze the data without these influential observations, it is best to create a new dataset.  

In this DATA step, a new dataset, PLANT_HEIGHT_2, is created. 

(21) The SET statement copies the data from PLANT_HEIGHT into PLANT_HEIGHT _2. 

(22) A new variable, ID, is created and each observation is assigned a number from 1 to N. 

(23) If the data is associated with ID #22, 102, or 146, it is deleted.  In other words, the 

influential observations are being deleted from the new dataset. 

(24) In this PROC PRINT, you should now see the new dataset, without those three influential 

data points.  Now, to conduct the analysis without the influential data points, simply re-run 

all of the above code using PLANT_HEIGHT_2 (instead of PLANT_HEIGHT) as your 

data. 

Note: The decision as to whether to exclude data points must be made carefully.  Such a 

decision should be based on a thorough knowledge of the data and theory.  This code is 

provided so that you have the option to exclude data and rerun analyses; it is not intended as a 

recommendation that you do so.  You must use your best judgment in that regard. 

Inferential Statistics 

Multiple Regression  

Selected Output 

Inferential Statistics 

Model Building / Variable Selection  

Research Scenario 

 

A transportation engineer, interested in designing safer roads, would like to know the best set of 

predictors for vehicular accidents (rate; expressed as the accident rate per million vehicle miles).  

Due to the high costs of data collection, he is not interested in the full model.  The predictors he 

is considering are: 

• average daily traffic count in thousands (adt), 

• truck volume as a percent of the total volume (trks), 

• speed limit in 1973 (slim), 

• total number of lanes of traffic (lane), and 

• number of access points per mile (acpt). 

 

This analysis will use the “Highway1.csv” file that can be found as a companion with this SAS 

guide.  If you prefer to enter the data manually, instead of reading in the data file, the raw data 

for the variables of interest may be found in the Appendix of this guide. 

 

 
 

Source of Data: Hoffstedt, C. (n.d.). [Highway accidents: Unpublished masters paper]. 

Unpublished raw data.  Retrieved from R Package “car.”  

Inferential Statistics 

Model Building / Variable Selection  

SAS Code 

 
DATA HIGHWAY; 

  INFILE "C:\Highway1.CSV" FIRSTOBS=2 DSD MISSOVER; 

  INPUT RATE LEN ADT TRKS SIGS1 SLIM SHLD LANE ACPT ITG  

   LWID HTYPE $; 

  KEEP RATE ADT TRKS SLIM LANE ACPT; 

RUN; 

 

PROC PRINT DATA=HIGHWAY; 

RUN; 

 

(1) PROC GLMSELECT DATA=HIGHWAY PLOTS=ALL; 

(2)  MODEL RATE=ADT TRKS SLIM LANE ACPT /  

(3)   SELECTION=STEPWISE  

(4)   DETAILS=STEPS  

(5)   SELECT=SL SLSTAY=0.05 SLENTRY=0.05; 

  TITLE "Stepwise Model Selection: Highway Accident Rate"; 

RUN; 

 

 PROC GLMSELECT DATA=HIGHWAY PLOTS=ALL; 

  MODEL RATE=ADT TRKS SLIM LANE ACPT /  

(6)   SELECTION=FORWARD  

   DETAILS=STEPS  

(7)   SELECT=SL SLENTRY=0.05; 

  TITLE "Forward Model Selection: Highway Accident Rate"; 

RUN; 

 

PROC GLMSELECT DATA=HIGHWAY PLOTS=ALL; 

  MODEL RATE=ADT TRKS SLIM LANE ACPT /  

(8)   SELECTION=BACKWARD  

   DETAILS=STEPS  

(9)   SELECT=SL SLSTAY=0.05; 

  TITLE "Backward Model Selection: Highway Accident Rate"; 

RUN; 

 

(10) PROC REG DATA=HIGHWAY PLOTS(ONLY)=(RSQUARE ADJRSQ CP); 

  MODEL RATE=ADT TRKS SLIM LANE ACPT /  

(11)   SELECTION=RSQUARE  

(12)   ADJRSQ CP; 

  TITLE "All Possible Subsets Analysis: Highway Accident Rate"; 

RUN; 

 

TITLE; 

QUIT; 

 

(1) PROC GLMSELECT is used for model building. 

(2) The MODEL statement is written as Dependent Variable (DV) = Independent Variable 

(IV).  Be sure to include all of the IVs that are candidates for inclusion. 

(3) The SELECTION option is used to identify the model building method you want to use 

(STEPWISE, FORWARD, or BACKWARD).  Here, STEPWISE was chosen. 



R 3.4.1: A Survival Guide 

 

151 

 

(4) The DETAILS=STEPS option is used if you want to see what SAS did at each step of 

model building (as opposed to seeing the final model only). 

(5) The SELECT= option is used to identify the statistic that is used in variable selection.  

Options include AIC, AICC, BIC, and SBC. Here, SL is used.  SL refers to “significance 

level.”  In other words, a variable has to make a significant p value contribution to the 

model.  If you choose SELECT=SL, you must also identify a significance level for a 

variable to qualify for entry in the model (SLENTRY=) and a significance level for a 

variable to qualify to stay in the model (SLSTAY=). 

(6) The SELECTION option is used to identify the model building method you want to use 

(STEPWISE, FORWARD, or BACKWARD).  Here, FORWARD was chosen. 

(7) For forward selection, if you choose SELECT=SL, you must also identify the significance 

level for a variable to qualify for entry in the model (SLENTRY=). 

(8) The SELECTION option is used to identify the model building method you want to use 

(STEPWISE, FORWARD, or BACKWARD).  Here, BACKWARD was chosen. 

(9) For backward selection, if you choose SELECT=SL, you must also identify the 

significance level for a variable to qualify to stay in the model (SLSTAY=). 

(10) This PROC REG is used to conduct an all-possible-subsets analysis. 

(11) The SELECTION=RSQUARE option tells SAS to analyze all possible subsets and rank 

them based on their R-square values.  You may also use adjusted R-square 

(SELECTION=ADJRSQ) or Mallow’s Cp (SELECTION=CP). 

(12) The ADJRSQ and CP options request these values to be output for each subset, even 

though they were not used as the ranking criteria.  Remember: You may copy and paste 

the output into Excel if you want to sort or manipulate the results.  

Inferential Statistics 

Model Building / Variable Selection  

Selected Output 

Appendix  

US Cereal Data 

 

CALORIES CARBO 

212.12121 15.15152 

212.12121 21.21212 

100.00000 16.00000 

146.66667 14.00000 

110.00000 11.00000 

173.33333 24.00000 

134.32836 22.38806 

134.32836 19.40299 

160.00000 16.00000 

88.00000 13.60000 

160.00000 17.33333 

220.00000 26.00000 

110.00000 12.00000 

110.00000 22.00000 

100.00000 21.00000 

110.00000 13.00000 

110.00000 12.00000 

220.00000 20.00000 

110.00000 21.00000 

133.33333 14.66667 

133.33333 24.00000 

110.00000 11.00000 

146.66667 18.66667 

125.00000 17.50000 

179.10448 17.91045 

179.10448 20.89552 

146.66667 17.33333 

113.63636 12.50000 

146.66667 20.00000 

113.63636 17.04545 

440.00000 68.00000 

363.63636 39.39394 

120.00000 12.00000 

146.66667 15.33333 

82.70677 10.52632 

186.66667 26.66667 

73.33333 14.00000 

149.25373 17.91045 

110.00000 12.00000 

238.80597 25.37313 

100.00000 15.00000 

179.10448 22.38806 

208.95522 31.34328 

260.00000 27.00000 

179.10448 16.41791 

100.00000 20.00000 

50.00000 13.00000 

200.00000 28.00000 

160.00000 18.66667 

200.00000 21.00000 

180.00000 30.00000 

97.34513 20.35398 

110.00000 22.00000 

134.32836 28.35821 

134.32836 29.85075 

146.66667 12.00000 

110.00000 16.00000 

110.00000 21.00000 

140.00000 15.00000 

100.00000 16.00000 

146.66667 28.00000 

110.00000 13.00000 

149.25373 25.37313 

100.00000 17.00000 

146.66667 21.33333 

Appendix  

Plant Height Data 

 

HEIGHT LOGHT TEMP RAIN 

28.000000 1.447158 10.80 1208 

26.600000 1.424882 24.50 3015 

0.300000 -0.522880 20.90 278 

1.600000 0.204120 19.90 598 

0.200000 -0.698970 9.70 976 

1.700000 0.230449 22.60 1387 

0.500000 -0.301030 16.80 1283 

10.000000 1.000000 27.70 2585 

40.000000 1.602060 15.50 1262 

0.500000 -0.301030 26.40 1704 

0.550000 -0.259640 5.40 664 

32.000000 1.505150 25.20 2087 

5.000000 0.698970 19.30 3191 

7.000000 0.845098 27.20 3031 

12.000000 1.079181 24.80 2770 

1.680000 0.225309 15.30 355 

0.700000 -0.154900 20.50 926 

4.000000 0.602060 24.90 1831 

0.600000 -0.221850 23.50 2814 

1.600000 0.204120 13.80 598 

32.000000 1.505150 26.00 2110 

61.000000 1.785330 5.00 1427 

14.800000 1.170262 25.80 2012 

1.000000 0.000000 19.90 338 

15.000000 1.176091 24.90 2767 

7.000000 0.845098 26.80 1184 

20.000000 1.301030 25.00 2664 

7.000000 0.845098 25.40 2494 

2.900000 0.462398 26.30 2607 

9.670000 0.985426 18.20 3042 

8.000000 0.903090 24.10 2314 

2.000000 0.301030 22.30 216 

0.600000 -0.221850 15.70 1052 

1.700000 0.230449 13.00 723 

7.000000 0.845098 18.30 762 

0.080000 -1.096910 -11.10 252 

0.200000 -0.698970 14.90 1150 

0.707000 -0.150580 3.40 637 

7.000000 0.845098 20.70 703 

0.500000 -0.301030 13.00 214 

23.500000 1.371068 26.50 2542 

16.000000 1.204120 23.40 1315 

0.233000 -0.632640 0.20 526 

34.000000 1.531479 25.90 2315 

15.000000 1.176091 25.50 2462 

15.000000 1.176091 24.60 2660 

1.500000 0.176091 17.00 1003 

16.000000 1.204120 26.00 3991 

2.800000 0.447158 15.60 384 

25.000000 1.397940 25.90 3048 

20.000000 1.301030 27.30 1505 

0.400000 -0.397940 23.90 1027 

6.000000 0.778151 4.20 2043 

0.200000 -0.698970 -5.70 305 

3.500000 0.544068 4.80 599 

25.000000 1.397940 17.00 1307 

18.000000 1.255273 2.40 2142 

10.000000 1.000000 16.20 276 

10.000000 1.000000 25.20 2576 

0.500000 -0.301030 15.70 281 

19.000000 1.278754 24.90 3273 

20.000000 1.301030 24.60 2674 

3.000000 0.477121 17.90 583 

0.110000 -0.958610 13.70 630 

12.500000 1.096910 26.20 790 

0.600000 -0.221850 16.50 546 

0.050000 -1.301030 -5.10 257 

30.000000 1.477121 22.70 2726 

35.000000 1.544068 24.80 1649 

3.000000 0.477121 22.50 2865 

0.200000 -0.698970 1.20 788 

0.050000 -1.301030 3.70 500 

0.032200 -1.492140 16.80 420 

3.500000 0.544068 21.20 1741 

12.000000 1.079181 25.40 2494 

0.800000 -0.096910 -2.90 301 

16.000000 1.204120 13.80 603 

16.000000 1.204120 23.90 1036 

30.000000 1.477121 17.00 706 

30.000000 1.477121 26.50 1661 

18.100000 1.257679 26.00 3612 

10.000000 1.000000 7.00 508 

3.000000 0.477121 14.90 682 

7.000000 0.845098 19.40 964 

4.000000 0.602060 24.80 3283 

0.080000 -1.096910 -1.00 436 

5.000000 0.698970 5.00 1427 

2.500000 0.397940 24.30 2567 

32.000000 1.505150 23.00 1327 

30.000000 1.477121 9.90 1037 

10.000000 1.000000 27.10 1664 

2.000000 0.301030 9.00 781 

29.300000 1.466868 9.40 1975 

4.000000 0.602060 25.30 2444 

0.220000 -0.657580 -5.10 257 

41.000000 1.612784 9.50 2561 

24.000000 1.380211 19.80 793 

2.000000 0.301030 25.00 2662 

4.500000 0.653213 18.60 882 

2.400000 0.380211 16.90 1313 

2.000000 0.301030 27.30 1505 

0.040000 -1.397940 10.60 1936 

28.000000 1.447158 24.80 3283 

0.070000 -1.154900 -6.40 244 

0.280000 -0.552840 8.50 996 

5.000000 0.698970 24.80 2803 

0.500000 -0.301030 18.60 1099 

0.800000 -0.096910 -0.70 422 

35.000000 1.544068 7.00 972 

20.000000 1.301030 24.80 2993 

1.800000 0.255273 13.50 915 

19.000000 1.278754 12.70 1121 

0.350000 -0.455930 10.90 208 

0.250000 -0.602060 13.00 214 

30.000000 1.477121 7.50 1720 

15.000000 1.176091 22.70 1397 

10.000000 1.000000 25.10 2598 

3.000000 0.477121 22.90 73 

30.000000 1.477121 24.90 3329 

0.080000 -1.096910 -10.50 236 

2.020000 0.305351 20.20 475 

0.800000 -0.096910 23.90 1545 

1.150000 0.060698 12.40 1263 

0.450000 -0.346790 18.90 293 

6.000000 0.778151 24.80 3269 

0.150000 -0.823910 -1.20 657 

1.584893 0.200000 5.10 691 

0.140000 -0.853870 3.70 500 

5.000000 0.698970 20.90 278 

2.500000 0.397940 24.80 2920 

3.000000 0.477121 16.40 501 

20.000000 1.301030 15.30 780 

6.000000 0.778151 22.60 803 

1.700000 0.230449 16.70 484 

0.200000 -0.698970 6.40 1379 

6.000000 0.778151 16.20 272 

3.800000 0.579784 20.80 1698 

8.000000 0.903090 18.90 380 

9.000000 0.954243 19.10 1085 

0.600000 -0.221850 9.60 174 

0.239000 -0.621600 13.50 867 

4.500000 0.653213 21.00 834 

12.000000 1.079181 24.80 2835 

1.700000 0.230449 20.50 354 

0.810000 -0.091510 1.30 539 

13.500000 1.130334 -4.30 296 

0.500000 -0.301030 4.90 977 

0.720000 -0.142670 16.70 290 

1.500000 0.176091 17.50 1165 

1.710000 0.232996 20.10 520 

0.300000 -0.522880 13.70 1019 

3.000000 0.477121 7.90 1156 

8.000000 0.903090 27.50 1663 

2.900000 0.462398 18.10 597 

13.000000 1.113943 13.60 1016 

0.200000 -0.698970 4.30 374 

1.000000 0.000000 12.00 872 

39.600000 1.597695 26.50 1974 

0.158000 -0.801340 4.50 597 

0.500000 -0.301030 20.40 310 

9.000000 0.954243 20.40 310 

3.000000 0.477121 10.10 1176 

1.050000 0.021189 -2.10 1418 

0.500000 -0.301030 9.10 2421 

11.000000 1.041393 21.00 1476 

39.000000 1.591065 8.00 692 

1.940000 0.287802 16.60 212 

12.400000 1.093422 6.20 564 

1.500000 0.176091 7.80 1211 

1.000000 0.000000 15.30 656 

0.750000 -0.124940 27.00 2319 

4.000000 0.602060 12.10 859 

15.000000 1.176091 25.00 2616 

0.550000 -0.259640 26.00 1117 

6.000000 0.778151 24.90 2731 

0.500000 -0.301030 16.70 630 

15.000000 1.176091 2.70 572 

0.246000 -0.609060 3.50 1555 

Appendix  

Highway1 Data 

 

RATE ADT TRKS SLIM LANE ACPT 

4.58 69 8 55 8 4.6 

2.86 73 8 60 4 4.4 

3.02 49 10 60 4 4.7 

2.29 61 13 65 6 3.8 

1.61 28 12 70 4 2.2 

6.87 30 6 55 4 24.8 

3.85 46 8 55 4 11.0 

6.12 25 9 55 4 18.5 

3.29 43 12 50 4 7.5 

5.88 23 7 50 4 8.2 

4.20 23 6 60 4 5.4 

4.61 20 9 50 4 11.2 

4.80 18 14 50 2 15.2 

3.85 21 8 60 4 5.4 

2.69 27 7 55 4 7.9 

1.99 22 9 60 4 3.2 

2.01 19 9 60 4 11.0 

4.22 9 11 50 2 8.9 

2.76 12 8 55 2 12.4 

2.55 12 7 60 4 7.8 

1.89 15 13 55 4 9.6 

2.34 8 8 60 2 4.3 

2.83 5 9 50 2 11.1 

1.81 5 15 60 2 6.8 

9.23 23 6 40 4 53.0 

8.60 13 6 45 2 17.3 

8.21 7 8 55 2 27.3 

2.93 10 10 55 2 18.0 

7.48 12 7 45 2 30.2 

2.57 9 8 60 2 10.3 

5.77 4 8 45 2 18.2 

2.90 5 10 55 2 12.3 

2.97 4 13 55 2 7.1 

1.84 5 12 55 2 14.0 

3.78 2 10 55 2 11.3 

2.76 3 8 50 2 16.3 

4.27 1 11 55 2 9.6 

3.05 3 11 60 2 9.0 

4.12 1 10 55 2 10.4 

